

**var** — Vector autoregressive models<sup>+</sup>

<sup>+</sup>This command includes features that are part of [StataNow](#).

|                                |                                      |                                |                                      |
|--------------------------------|--------------------------------------|--------------------------------|--------------------------------------|
| <a href="#">Description</a>    | <a href="#">Quick start</a>          | <a href="#">Menu</a>           | <a href="#">Syntax</a>               |
| <a href="#">Options</a>        | <a href="#">Remarks and examples</a> | <a href="#">Stored results</a> | <a href="#">Methods and formulas</a> |
| <a href="#">Acknowledgment</a> | <a href="#">References</a>           | <a href="#">Also see</a>       |                                      |

## Description

`var` fits a multivariate time-series regression of each dependent variable on lags of itself and on lags of all the other dependent variables. `var` also fits a variant of vector autoregressive (VAR) models known as the VARX model, which also includes exogenous variables. See [\[TS\] var intro](#) for a list of commands that are used in conjunction with `var`.

## Quick start

Vector autoregressive model for dependent variables `y1`, `y2`, and `y3` and their first and second lags using `tsset` data

```
var y1 y2 y3
```

Same as above, but include second and third lags instead of first and second

```
var y1 y2 y3, lags(2 3)
```

Add exogenous variables `x1` and `x2`

```
var y1 y2 y3, lags(2 3) exog(x1 x2)
```

Same as above, but make a small-sample degrees-of-freedom adjustment

```
var y1 y2 y3, lags(2 3) exog(x1 x2) dfk
```

## Menu

Statistics > Multivariate time series > Vector autoregression (VAR)

## Syntax

```
var depvarlist [if] [in] [, options]
```

| <i>options</i>                           | Description  |
|--|--|
| Model                                    |  |
| <code>noconstant</code>                  | suppress constant term   |
| <code>lags(<i>numlist</i>)</code>        | specify a list of lags for the VAR model   |
| <code>exog(<i>varlist</i>)</code>        | specify exogenous variables  |
| Model 2                                  |  |
| <code>constraints(<i>numlist</i>)</code> | apply specified linear constraints   |
| <code>[no]log</code>                     | display or suppress SURE iteration log; default is to display                    |
| <code>iterate(#)</code>                  | set maximum number of iterations for SURE; default is <code>iterate(1600)</code> |
| <code>tolerance(#)</code>                | set convergence tolerance of SURE  |
| <code>noisure</code>                     | use one-step SURE  |
| <code>dfk</code>                         | make small-sample degrees-of-freedom adjustment                                  |
| <code>small</code>                       | report small-sample <i>t</i> and <i>F</i> statistics                             |
| <code>nobigf</code>                      | do not compute parameter vector for coefficients implicitly set to zero          |
| SE/Robust                                |  |
| <code>+vce(<i>vcetype</i>)</code>        | <i>vcetype</i> may be <code>unadjusted</code> or <code>robust</code>             |
| Reporting                                |  |
| <code>level(#)</code>                    | set confidence level; default is <code>level(95)</code>                          |
| <code>lutstats</code>                    | report Lütkepohl lag-order selection statistics                                  |
| <code>nocnsreport</code>                 | do not display constraints   |
| <code>display_options</code>             | control columns and column formats, row spacing, and line width                  |
| <code>coeflegend</code>                  | display legend instead of statistics   |

<sup>+</sup>These features are part of [StataNow](#).

You must `tsset` your data before using `var`; see [\[TS\] tsset](#).

*depvarlist* and *varlist* may contain time-series operators; see [\[U\] 11.4.4 Time-series varlists](#).

`bayes`, `by`, `collect`, `fp`, `rolling`, `statsby`, and `xi` are allowed; see [\[U\] 11.1.10 Prefix commands](#).

For more details, see [\[BAYES\] bayes: var](#).

`coeflegend` does not appear in the dialog box.

See [\[U\] 20 Estimation and postestimation commands](#) for more capabilities of estimation commands.

## Options

Model

`noconstant`; see [\[R\] Estimation options](#).

`lags(numlist)` specifies the lags to be included in the model. The default is `lags(1 2)`. This option takes a *numlist* and not simply an integer for the maximum lag. For example, `lags(2)` would include only the second lag in the model, whereas `lags(1/2)` would include both the first and second lags in the model. See [\[U\] 11.1.8 numlist](#) and [\[U\] 11.4.4 Time-series varlists](#) for more discussion of *numlists* and lags.

`exog(varlist)` specifies a list of exogenous variables to be included in the VAR model.

---

 Model 2
 

---

`constraints(numlist)`; see [R] [Estimation options](#).

`log` and `nolog` specify whether to display the log from the iterated seemingly unrelated regression algorithm. By default, the iteration log is displayed when the coefficients are estimated through iterated seemingly unrelated regression. When the `constraints()` option is not specified, the estimates are obtained via OLS, and `nolog` has no effect. For this reason, `log` and `nolog` can be specified only when `constraints()` is specified. Similarly, `nolog` cannot be combined with `noisure`. You can also control the iteration log using `set iterlog`; see [R] [set iter](#).

`iterate(#)` specifies an integer that sets the maximum number of iterations when the estimates are obtained through iterated seemingly unrelated regression. By default, the limit is 1,600. When `constraints()` is not specified, the estimates are obtained using OLS, and `iterate()` has no effect. For this reason, `iterate()` can be specified only when `constraints()` is specified. Similarly, `iterate()` cannot be combined with `noisure`.

`tolerance(#)` specifies a number greater than zero and less than 1 for the convergence tolerance of the iterated seemingly unrelated regression algorithm. By default, the tolerance is  $1e-6$ . When the `constraints()` option is not specified, the estimates are obtained using OLS, and `tolerance()` has no effect. For this reason, `tolerance()` can be specified only when `constraints()` is specified. Similarly, `tolerance()` cannot be combined with `noisure`.

`noisure` specifies that the estimates in the presence of constraints be obtained through one-step seemingly unrelated regression. By default, `var` obtains estimates in the presence of constraints through iterated seemingly unrelated regression. When `constraints()` is not specified, the estimates are obtained using OLS, and `noisure` has no effect. For this reason, `noisure` can be specified only when `constraints()` is specified.

`dfk` specifies that a small-sample degrees-of-freedom adjustment be used when estimating  $\Sigma$ , the error variance–covariance matrix. Specifically,  $1/(T - \bar{m})$  is used instead of the large-sample divisor  $1/T$ , where  $\bar{m}$  is the average number of parameters in the functional form for  $y_t$  over the  $K$  equations.

`small` causes `var` to report small-sample  $t$  and  $F$  statistics instead of the large-sample normal and  $\chi^2$  statistics.

`nobifg` requests that `var` not save the estimated parameter vector that incorporates coefficients that have been implicitly constrained to be zero, such as when some lags have been omitted from a model. `e(bf)` is used for computing asymptotic standard errors in the postestimation commands `irf create` and `fcast compute`; see [TS] [irf create](#) and [TS] [fcast compute](#). Therefore, specifying `nobifg` implies that the asymptotic standard errors will not be available from `irf create` and `fcast compute`. See [Fitting models with some lags excluded](#).

---

 SE/Robust
 

---

`vce(vcetype)` is part of `StataNow`. It specifies the type of standard error reported, which includes types that are derived from asymptotic theory (`unadjusted`) and that are robust to some kinds of misspecification (`robust`); see [R] [vce\\_option](#).

`vce(unadjusted)`, the default, specifies that an unadjusted (nonrobust) VCE matrix be used; this results in efficient estimates when assuming homoskedasticity.

Reporting

level(#); see [R] [Estimation options](#).

lutstats specifies that the Lütkepohl (2005) versions of the lag-order selection statistics be reported.

See [Methods and formulas](#) in [TS] [varso](#) for a discussion of these statistics.

nocnsreport; see [R] [Estimation options](#).

*display\_options*: noci, nopvalues, vsquish, cformat(%fmt), pformat(%fmt), sformat(%fmt), and nolstretch; see [R] [Estimation options](#).

The following option is available with var but is not shown in the dialog box:

coeflegend; see [R] [Estimation options](#).

## Remarks and examples

[stata.com](http://www.stata.com)

Remarks are presented under the following headings:

*Introduction*

*Fitting models with some lags excluded*

*Fitting models with exogenous variables*

*Fitting models with constraints on the coefficients*

## Introduction

A VAR model is a model in which  $K$  variables are specified as linear functions of  $p$  of their own lags,  $p$  lags of the other  $K - 1$  variables, and possibly exogenous variables. A VAR model with  $p$  lags is usually denoted a VAR( $p$ ). For more information, see [TS] [var intro](#).

### ▶ Example 1: VAR model

To illustrate the basic usage of var, we replicate the example in Lütkepohl (2005, 77–78). The data consists of three variables: the first difference of the natural log of investment, dln\_inv; the first difference of the natural log of income, dln\_inc; and the first difference of the natural log of consumption, dln\_consump. The dataset contains data through the fourth quarter of 1982, though Lütkepohl uses only the observations through the fourth quarter of 1978.

```
. use https://www.stata-press.com/data/r18/lutkepohl2
(Quarterly SA West German macro data, Bil DM, from Lutkepohl 1993 Table E.1)
. tsset
Time variable: qtr, 1960q1 to 1982q4
Delta: 1 quarter
. var dln_inv dln_inc dln_consump if qtr<=tq(1978q4), lutstats dfk
Vector autoregression
Sample: 1960q4 thru 1978q4
Log likelihood = 606.307 (lutstats) AIC = -24.63163
FPE = 2.18e-11 HQIC = -24.40656
Det(Sigma_ml) = 1.23e-11 SBIC = -24.06686
```

| Equation    | Parms | RMSE    | R-sq   | chi2     | P>chi2 |
|-------------|-------|---------|--------|----------|--------|
| dln_inv     | 7     | .046148 | 0.1286 | 9.736909 | 0.1362 |
| dln_inc     | 7     | .011719 | 0.1142 | 8.508289 | 0.2032 |
| dln_consump | 7     | .009445 | 0.2513 | 22.15096 | 0.0011 |

|             | Coefficient | Std. err. | z     | P> z  | [95% conf. interval] |           |
|-------------|-------------|-----------|-------|-------|----------------------|-----------|
| dln_inv     |             |           |       |       |                      |           |
| dln_inv     |             |           |       |       |                      |           |
| L1.         | -.3196318   | .1254564  | -2.55 | 0.011 | -.5655218            | -.0737419 |
| L2.         | -.1605508   | .1249066  | -1.29 | 0.199 | -.4053633            | .0842616  |
| dln_inc     |             |           |       |       |                      |           |
| L1.         | .1459851    | .5456664  | 0.27  | 0.789 | -.9235013            | 1.215472  |
| L2.         | .1146009    | .5345709  | 0.21  | 0.830 | -.9331388            | 1.162341  |
| dln_consump |             |           |       |       |                      |           |
| L1.         | .9612288    | .6643086  | 1.45  | 0.148 | -.3407922            | 2.26325   |
| L2.         | .9344001    | .6650949  | 1.40  | 0.160 | -.369162             | 2.237962  |
| _cons       | -.0167221   | .0172264  | -0.97 | 0.332 | -.0504852            | .0170409  |
| dln_inc     |             |           |       |       |                      |           |
| dln_inv     |             |           |       |       |                      |           |
| L1.         | .0439309    | .0318592  | 1.38  | 0.168 | -.018512             | .1063739  |
| L2.         | .0500302    | .0317196  | 1.58  | 0.115 | -.0121391            | .1121995  |
| dln_inc     |             |           |       |       |                      |           |
| L1.         | -.1527311   | .1385702  | -1.10 | 0.270 | -.4243237            | .1188615  |
| L2.         | .0191634    | .1357525  | 0.14  | 0.888 | -.2469067            | .2852334  |
| dln_consump |             |           |       |       |                      |           |
| L1.         | .2884992    | .168699   | 1.71  | 0.087 | -.0421448            | .6191431  |
| L2.         | -.0102      | .1688987  | -0.06 | 0.952 | -.3412354            | .3208353  |
| _cons       | .0157672    | .0043746  | 3.60  | 0.000 | .0071932             | .0243412  |
| dln_consump |             |           |       |       |                      |           |
| dln_inv     |             |           |       |       |                      |           |
| L1.         | -.002423    | .0256763  | -0.09 | 0.925 | -.0527476            | .0479016  |
| L2.         | .0338806    | .0255638  | 1.33  | 0.185 | -.0162235            | .0839847  |
| dln_inc     |             |           |       |       |                      |           |
| L1.         | .2248134    | .1116778  | 2.01  | 0.044 | .005929              | .4436978  |
| L2.         | .3549135    | .1094069  | 3.24  | 0.001 | .1404798             | .5693471  |
| dln_consump |             |           |       |       |                      |           |
| L1.         | -.2639695   | .1359595  | -1.94 | 0.052 | -.5304451            | .0025062  |
| L2.         | -.0222264   | .1361204  | -0.16 | 0.870 | -.2890175            | .2445646  |
| _cons       | .0129258    | .0035256  | 3.67  | 0.000 | .0060157             | .0198358  |

Note: Small-sample degrees-of-freedom adjustment applied when estimating covariance matrix of residuals.

The output has two parts: a header and the standard Stata output table for the coefficients, standard errors, and confidence intervals. The header contains summary statistics for each equation in the VAR model and statistics used in selecting the lag order of the VAR model. Although there are standard formulas for all the lag-order statistics, Lütkepohl (2005) gives different versions of the three information criteria that drop the constant term from the likelihood. To obtain the Lütkepohl (2005) versions, we specified the `lutstats` option. The formulas for the standard and Lütkepohl versions of these statistics are given in *Methods and formulas* of [TS] `varsoc`.

The `dfk` option specifies that the small-sample divisor  $1/(T - \bar{m})$  be used in estimating  $\Sigma$  instead of the maximum likelihood (ML) divisor  $1/T$ , where  $\bar{m}$  is the average number of parameters included in each of the  $K$  equations. All the lag-order statistics are computed using the ML estimator of  $\Sigma$ . Thus, specifying `dfk` will not change the computed lag-order statistics, but it will change the estimated variance–covariance matrix. Also, when `dfk` is specified, a `dfk`-adjusted log likelihood is computed and stored in `e(ll_dfk)`.

◀

The `lag()` option takes a *numlist* of lags. To specify a model that includes the first and second lags, type

```
. var y1 y2 y3, lags(1/2)
```

not

```
. var y1 y2 y3, lags(2)
```

because the latter specification would fit a model that included only the second lag.

## Fitting models with some lags excluded

To fit a model that has only a fourth lag, that is,

$$\mathbf{y}_t = \mathbf{v} + \mathbf{A}_4 \mathbf{y}_{t-4} + \mathbf{u}_t$$

you would specify the `lags(4)` option. Doing so is equivalent to fitting the more general model

$$\mathbf{y}_t = \mathbf{v} + \mathbf{A}_1 \mathbf{y}_{t-1} + \mathbf{A}_2 \mathbf{y}_{t-2} + \mathbf{A}_3 \mathbf{y}_{t-3} + \mathbf{A}_4 \mathbf{y}_{t-4} + \mathbf{u}_t$$

with  $\mathbf{A}_1$ ,  $\mathbf{A}_2$ , and  $\mathbf{A}_3$  constrained to be  $\mathbf{0}$ . When you fit a model with some lags excluded, `var` estimates the coefficients included in the specification ( $\mathbf{A}_4$  here) and stores these estimates in `e(b)`. To obtain the asymptotic standard errors for impulse–response functions and other postestimation statistics, Stata needs the complete set of parameter estimates, including those that are constrained to be zero; `var` stores them in `e(bf)`. Because you can specify models for which the full set of parameter estimates exceeds Stata’s limit on the size of matrices, the `nobigf` option specifies that `var` not compute and store `e(bf)`. This means that the asymptotic standard errors of the postestimation functions cannot be obtained, although bootstrap standard errors are still available. Building `e(bf)` can be time consuming, so if you do not need this full matrix, and speed is an issue, use `nobigf`.

## Fitting models with exogenous variables

### ▷ Example 2: VAR model with exogenous variables

We use the `exog()` option to include exogenous variables in a VAR model.

```
. var dln_inc dln_consump if qtr<=tq(1978q4), dfk exog(dln_inv)
Vector autoregression
Sample: 1960q4 thru 1978q4      Number of obs   =      73
Log likelihood = 478.5663      AIC              = -12.78264
FPE            = 9.64e-09      HQIC             = -12.63259
Det(Sigma_ml) = 6.93e-09      SBIC             = -12.40612
```

| Equation    | Parms | RMSE    | R-sq   | chi2     | P>chi2 |
|-------------|-------|---------|--------|----------|--------|
| dln_inc     | 6     | .011917 | 0.0702 | 5.059587 | 0.4087 |
| dln_consump | 6     | .009197 | 0.2794 | 25.97262 | 0.0001 |

|             | Coefficient | Std. err. | z     | P> z  | [95% conf. interval] |           |
|-------------|-------------|-----------|-------|-------|----------------------|-----------|
| dln_inc     |             |           |       |       |                      |           |
| dln_inc     |             |           |       |       |                      |           |
| L1.         | -.1343345   | .1391074  | -0.97 | 0.334 | -.4069801            | .1383111  |
| L2.         | .0120331    | .1380346  | 0.09  | 0.931 | -.2585097            | .2825759  |
| dln_consump |             |           |       |       |                      |           |
| L1.         | .3235342    | .1652769  | 1.96  | 0.050 | -.0004027            | .647471   |
| L2.         | .0754177    | .1648624  | 0.46  | 0.647 | -.2477066            | .398542   |
| dln_inv     | .0151546    | .0302319  | 0.50  | 0.616 | -.0440987            | .074408   |
| _cons       | .0145136    | .0043815  | 3.31  | 0.001 | .0059259             | .0231012  |
| dln_consump |             |           |       |       |                      |           |
| dln_inc     |             |           |       |       |                      |           |
| L1.         | .2425719    | .1073561  | 2.26  | 0.024 | .0321578             | .452986   |
| L2.         | .3487949    | .1065281  | 3.27  | 0.001 | .1400036             | .5575862  |
| dln_consump |             |           |       |       |                      |           |
| L1.         | -.3119629   | .1275524  | -2.45 | 0.014 | -.5619611            | -.0619648 |
| L2.         | -.0128502   | .1272325  | -0.10 | 0.920 | -.2622213            | .2365209  |
| dln_inv     | .0503616    | .0233314  | 2.16  | 0.031 | .0046329             | .0960904  |
| _cons       | .0131013    | .0033814  | 3.87  | 0.000 | .0064738             | .0197288  |

Note: Small-sample degrees-of-freedom adjustment applied when estimating covariance matrix of residuals.

All the postestimation commands for analyzing VAR models work when exogenous variables are included in a model, but the asymptotic standard errors for the  $h$ -step-ahead forecasts are not available.

## Fitting models with constraints on the coefficients

`var` permits model specifications that include constraints on the coefficient, though `var` does not allow for constraints on  $\Sigma$ . See [TS] [var intro](#) and [TS] [var svar](#) for ways to constrain  $\Sigma$ .

### ► Example 3: VAR model with constraints

In the [first example](#), we fit a full VAR(2) model to a three-equation model. The coefficients in the equation for `dln_inv` were jointly insignificant, as were the coefficients in the equation for `dln_inc`; and many individual coefficients were not significantly different from zero. In this example, we constrain the coefficient on `L2.dln_inc` in the equation for `dln_inv` and the coefficient on `L2.dln_consump` in the equation for `dln_inc` to be zero.

```
. constraint 1 [dln_inv]L2.dln_inc = 0
. constraint 2 [dln_inc]L2.dln_consump = 0
. var dln_inv dln_inc dln_consump if qtr<=tq(1978q4), lutstats dfk
> constraints(1 2)
Estimating VAR coefficients
Iteration 1:  tolerance = .00737681
Iteration 2:  tolerance = 3.998e-06
Iteration 3:  tolerance = 2.730e-09

Vector autoregression
Sample: 1960q4 thru 1978q4                Number of obs   =          73
Log likelihood = 606.2804                (lutstats) AIC   = -31.69254
FPE           = 1.77e-14                  HQIC           = -31.46747
Det(Sigma_ml) = 1.05e-14                  SBIC           = -31.12777

Equation      Parms      RMSE      R-sq      chi2      P>chi2
-----
dln_inv       6          .043895   0.1280    9.842338   0.0798
dln_inc       6          .011143   0.1141    8.584446   0.1268
dln_consump   7          .008981   0.2512    22.86958   0.0008

( 1)  [dln_inv]L2.dln_inc = 0
( 2)  [dln_inc]L2.dln_consump = 0
```



|                    | Coefficient | Std. err. | z     | P> z  | [95% conf. interval] |           |
|--------------------|-------------|-----------|-------|-------|----------------------|-----------|
| <b>dln_inv</b>     |             |           |       |       |                      |           |
| <b>dln_inv</b>     |             |           |       |       |                      |           |
| L1.                | -.320713    | .1247512  | -2.57 | 0.010 | -.5652208            | -.0762051 |
| L2.                | -.1607084   | .124261   | -1.29 | 0.196 | -.4042555            | .0828386  |
| <b>dln_inc</b>     |             |           |       |       |                      |           |
| L1.                | .1195448    | .5295669  | 0.23  | 0.821 | -.9183873            | 1.157477  |
| L2.                | 5.66e-19    | 9.33e-18  | 0.06  | 0.952 | -1.77e-17            | 1.89e-17  |
| <b>dln_consump</b> |             |           |       |       |                      |           |
| L1.                | 1.009281    | .623501   | 1.62  | 0.106 | -.2127586            | 2.231321  |
| L2.                | 1.008079    | .5713486  | 1.76  | 0.078 | -.1117438            | 2.127902  |
| _cons              | -.0162102   | .016893   | -0.96 | 0.337 | -.0493199            | .0168995  |
| <b>dln_inc</b>     |             |           |       |       |                      |           |
| <b>dln_inv</b>     |             |           |       |       |                      |           |
| L1.                | .0435712    | .0309078  | 1.41  | 0.159 | -.017007             | .1041495  |
| L2.                | .0496788    | .0306455  | 1.62  | 0.105 | -.0103852            | .1097428  |
| <b>dln_inc</b>     |             |           |       |       |                      |           |
| L1.                | -.1555119   | .1315854  | -1.18 | 0.237 | -.4134146            | .1023908  |
| L2.                | .0122353    | .1165811  | 0.10  | 0.916 | -.2162595            | .2407301  |
| <b>dln_consump</b> |             |           |       |       |                      |           |
| L1.                | .29286      | .1568345  | 1.87  | 0.062 | -.01453              | .6002501  |
| L2.                | -1.53e-18   | 1.89e-17  | -0.08 | 0.935 | -3.85e-17            | 3.55e-17  |
| _cons              | .015689     | .003819   | 4.11  | 0.000 | .0082039             | .0231741  |
| <b>dln_consump</b> |             |           |       |       |                      |           |
| <b>dln_inv</b>     |             |           |       |       |                      |           |
| L1.                | -.0026229   | .0253538  | -0.10 | 0.918 | -.0523154            | .0470696  |
| L2.                | .0337245    | .0252113  | 1.34  | 0.181 | -.0156888            | .0831378  |
| <b>dln_inc</b>     |             |           |       |       |                      |           |
| L1.                | .2224798    | .1094349  | 2.03  | 0.042 | .0079912             | .4369683  |
| L2.                | .3469758    | .1006026  | 3.45  | 0.001 | .1497984             | .5441532  |
| <b>dln_consump</b> |             |           |       |       |                      |           |
| L1.                | -.2600227   | .1321622  | -1.97 | 0.049 | -.519056             | -.0009895 |
| L2.                | -.0146825   | .1117618  | -0.13 | 0.895 | -.2337315            | .2043666  |
| _cons              | .0129149    | .003376   | 3.83  | 0.000 | .0062981             | .0195317  |

Note: Small-sample degrees-of-freedom adjustment applied when estimating covariance matrix of residuals.

None of the free parameter estimates changed by much. Whereas the coefficients in the equation `dln_inv` are now significant at the 10% level, the coefficients in the equation for `dln_inc` remain jointly insignificant.

## Stored results

`var` stores the following in `e()`:

### Scalars

|                           |   |
|---------------------------|---|
| <code>e(N)</code>         | number of observations  |
| <code>e(N_gaps)</code>    | number of gaps in sample  |
| <code>e(k)</code>         | number of parameters  |
| <code>e(k_eq)</code>      | number of equations in <code>e(b)</code>                        |
| <code>e(k_dv)</code>      | number of dependent variables                                   |
| <code>e(df_eq)</code>     | average number of parameters in an equation                     |
| <code>e(df_m)</code>      | model degrees of freedom  |
| <code>e(df_r)</code>      | residual degrees of freedom ( <i>small</i> only)                |
| <code>e(ll)</code>        | log likelihood  |
| <code>e(ll_dfk)</code>    | <i>dfk</i> adjusted log likelihood ( <i>dfk</i> only)           |
| <code>e(obs_#)</code>     | number of observations on equation #                            |
| <code>e(k_#)</code>       | number of parameters in equation #                              |
| <code>e(df_m#)</code>     | model degrees of freedom for equation #                         |
| <code>e(df_r#)</code>     | residual degrees of freedom for equation # ( <i>small</i> only) |
| <code>e(r2_#)</code>      | $R^2$ for equation #  |
| <code>e(ll_#)</code>      | log likelihood for equation #                                   |
| <code>e(chi2_#)</code>    | $\chi^2$ for equation #   |
| <code>e(F_#)</code>       | $F$ statistic for equation # ( <i>small</i> only)               |
| <code>e(p_#)</code>       | $p$ -value for equation #                                       |
| <code>e(rmse_#)</code>    | root mean squared error for equation #                          |
| <code>e(aic)</code>       | Akaike information criterion                                    |
| <code>e(hqic)</code>      | Hannan–Quinn information criterion                              |
| <code>e(sbic)</code>      | Schwarz’s Bayesian information criterion                        |
| <code>e(fpe)</code>       | final prediction error  |
| <code>e(mlag)</code>      | highest lag in VAR model  |
| <code>e(tmin)</code>      | first time period in sample                                     |
| <code>e(tmax)</code>      | maximum time  |
| <code>e(detsig)</code>    | determinant of <code>e(Sigma)</code>                            |
| <code>e(detsig_ml)</code> | determinant of $\hat{\Sigma}_{ml}$                              |
| <code>e(rank)</code>      | rank of <code>e(V)</code>                                       |

### Macros

|                                |   |
|--------------------------------|---|
| <code>e(cmd)</code>            | <code>var</code>  |
| <code>e(cmdline)</code>        | command as typed  |
| <code>e(depvar)</code>         | names of dependent variables  |
| <code>e(endog)</code>          | names of endogenous variables, if specified                           |
| <code>e(exog)</code>           | names of exogenous variables, and their lags, if specified            |
| <code>e(exogvars)</code>       | names of exogenous variables, if specified                            |
| <code>e(eqnames)</code>        | names of equations  |
| <code>e(lags)</code>           | lags in model   |
| <code>e(exlags)</code>         | lags of exogenous variables in model, if specified                    |
| <code>e(title)</code>          | title in estimation output  |
| <code>e(nocons)</code>         | <code>nocons</code> , if <code>noconstant</code> is specified         |
| <code>e(vce)</code>            | <i>vce</i> type specified in <code>vce()</code>                       |
| <code>e(vcetype)</code>        | title used to label Std. err.   |
| <code>e(constraints)</code>    | <code>constraints</code> , if <code>constraints()</code> is specified |
| <code>e(cnslist_var)</code>    | list of specified constraints   |
| <code>e(small)</code>          | <i>small</i> , if specified   |
| <code>e(lutstats)</code>       | <code>lutstats</code> , if specified                                  |
| <code>e(timevar)</code>        | time variable specified in <code>tsset</code>                         |
| <code>e(tsfmt)</code>          | format for the current time variable                                  |
| <code>e(dfk)</code>            | <i>dfk</i> , if specified   |
| <code>e(properties)</code>     | <code>b V</code>  |
| <code>e(predict)</code>        | program used to implement <code>predict</code>                        |
| <code>e(marginsok)</code>      | predictions allowed by <code>margins</code>                           |
| <code>e(marginsnotok)</code>   | predictions disallowed by <code>margins</code>                        |
| <code>e(marginsdefault)</code> | default <code>predict()</code> specification for <code>margins</code> |

|            |   |
|------------|---|
| Matrices   |   |
| e(b)       | coefficient vector                            |
| e(Cns)     | constraints matrix                            |
| e(Sigma)   | $\widehat{\Sigma}$ matrix                     |
| e(V)       | variance–covariance matrix of the estimators  |
| e(bf)      | constrained coefficient vector                |
| e(exlagsm) | matrix mapping lags to exogenous variables    |
| e(G)       | Gamma matrix; see <i>Methods and formulas</i> |
| Functions  |   |
| e(sample)  | marks estimation sample                       |

In addition to the above, the following is stored in `r()`:

|          |   |
|----------|---|
| Matrices |   |
| r(table) | matrix containing the coefficients with their standard errors, test statistics, $p$ -values, and confidence intervals |

Note that results stored in `r()` are updated when the command is replayed and will be replaced when any `r`-class command is run after the estimation command.

## Methods and formulas

When there are no constraints placed on the coefficients, the VAR( $p$ ) model is a seemingly unrelated regression model with the same explanatory variables in each equation. As discussed in Lütkepohl (2005) and Greene (2008, 696), performing linear regression on each equation produces the maximum likelihood estimates of the coefficients. The estimated coefficients can then be used to calculate the residuals, which in turn are used to estimate the cross-equation error variance–covariance matrix  $\Sigma$ .

Per Lütkepohl (2005), we write the VAR( $p$ ) model with exogenous variables as

$$\mathbf{y}_t = \mathbf{A}\mathbf{Y}_{t-1} + \mathbf{B}_0\mathbf{x}_t + \mathbf{u}_t \quad (5)$$

where

$\mathbf{y}_t$  is the  $K \times 1$  vector of endogenous variables,

$\mathbf{A}$  is a  $K \times Kp$  matrix of coefficients,

$\mathbf{B}_0$  is a  $K \times M$  matrix of coefficients,

$\mathbf{x}_t$  is the  $M \times 1$  vector of exogenous variables,

$\mathbf{u}_t$  is the  $K \times 1$  vector of white noise innovations, and

$\mathbf{Y}_t$  is the  $Kp \times 1$  matrix given by  $\mathbf{Y}_t = \begin{pmatrix} \mathbf{y}_t \\ \vdots \\ \mathbf{y}_{t-p+1} \end{pmatrix}$

Although (5) is easier to read, the formulas are much easier to manipulate if it is instead written as

$$\mathbf{Y} = \mathbf{B}\mathbf{Z} + \mathbf{U}$$

where

$$\begin{aligned} \mathbf{Y} &= (\mathbf{y}_1, \dots, \mathbf{y}_T) & \mathbf{Y} & \text{is } K \times T \\ \mathbf{B} &= (\mathbf{A}, \mathbf{B}_0) & \mathbf{B} & \text{is } K \times (Kp + M) \\ \mathbf{Z} &= \begin{pmatrix} \mathbf{Y}_0 \dots, \mathbf{Y}_{T-1} \\ \mathbf{x}_1 \dots, \mathbf{x}_T \end{pmatrix} & \mathbf{Z} & \text{is } (Kp + M) \times T \\ \mathbf{U} &= (\mathbf{u}_1, \dots, \mathbf{u}_T) & \mathbf{U} & \text{is } K \times T \end{aligned}$$

Intercept terms in the model are included in  $\mathbf{x}_t$ . If there are no exogenous variables and no intercept terms in the model,  $\mathbf{x}_t$  is empty.

The coefficients are estimated by iterated seemingly unrelated regression. Because the estimation is actually performed by `reg3`, the methods are documented in [R] [reg3](#). See [P] [makecns](#) for more on estimation with constraints.

Let  $\widehat{\mathbf{U}}$  be the matrix of residuals that are obtained via  $\mathbf{Y} - \widehat{\mathbf{B}}\mathbf{Z}$ , where  $\widehat{\mathbf{B}}$  is the matrix of estimated coefficients. Then the estimator of  $\mathbf{\Sigma}$  is

$$\widehat{\mathbf{\Sigma}} = \frac{1}{\widetilde{T}} \widehat{\mathbf{U}}' \widehat{\mathbf{U}}$$

By default, the maximum likelihood divisor of  $\widetilde{T} = T$  is used. When `dfk` is specified, a small-sample degrees-of-freedom adjustment is used; then,  $\widetilde{T} = T - \overline{m}$  where  $\overline{m}$  is the average number of parameters per equation in the functional form for  $\mathbf{y}_t$  over the  $K$  equations.

`small` specifies that Wald tests after `var` be assumed to have  $F$  or  $t$  distributions instead of  $\chi^2$  or standard normal distributions. The standard errors from each equation are computed using the degrees of freedom for the equation.

The “gamma” matrix stored in `e(G)` referred to in [Stored results](#) is the  $(Kp + 1) \times (Kp + 1)$  matrix given by

$$\frac{1}{T} \sum_{t=1}^T (\mathbf{1}, \mathbf{Y}'_t)(\mathbf{1}, \mathbf{Y}'_t)'$$

The formulas for the lag-order selection criteria and the log likelihood are discussed in [TS] [varsoc](#).

In [StataNow](#), this command supports the Huber/White/sandwich estimator of the variance using `vce(robust)`. See [P] [\\_robust](#), particularly [Introduction](#) and [Methods and formulas](#).

## Acknowledgment

We thank Christopher F. Baum of the Department of Economics at Boston College and author of the Stata Press books [An Introduction to Modern Econometrics Using Stata](#) and [An Introduction to Stata Programming](#) and coauthor of the Stata Press book [Environmental Econometrics Using Stata](#) for his helpful comments.

## References

- Abrego, M. R. M., and I. Love. 2016. Estimation of panel vector autoregression in Stata. *Stata Journal* 16: 778–804.
- Baum, C. F., and S. Hurn. 2021. *Environmental Econometrics Using Stata*. College Station, TX: Stata Press.
- Box-Steffensmeier, J. M., J. R. Freeman, M. P. Hitt, and J. C. W. Pevehouse. 2014. *Time Series Analysis for the Social Sciences*. New York: Cambridge University Press.
- Greene, W. H. 2008. *Econometric Analysis*. 6th ed. Upper Saddle River, NJ: Prentice Hall.
- Hamilton, J. D. 1994. *Time Series Analysis*. Princeton, NJ: Princeton University Press.
- Lütkepohl, H. 1993. *Introduction to Multiple Time Series Analysis*. 2nd ed. New York: Springer.
- . 2005. *New Introduction to Multiple Time Series Analysis*. New York: Springer.
- Rajbhandari, A. 2016. Vector autoregression–simulation, estimation, and inference in Stata. *The Stata Blog: Not Elsewhere Classified*. <https://blog.stata.com/2016/02/18/vector-autoregressionsimulation-estimation-and-inference-in-stata/>.

- Rossi, B., and Y. Wang. 2019. Vector autoregressive-based Granger causality test in the presence of instabilities. *Stata Journal* 19: 883–899.
- Schenck, D. 2016. Vector autoregressions in Stata. *The Stata Blog: Not Elsewhere Classified*. <https://blog.stata.com/2016/08/09/vector-autoregressions-in-stata/>.
- Stock, J. H., and M. W. Watson. 2001. Vector autoregressions. *Journal of Economic Perspectives* 15: 101–115. <https://doi.org/10.1257/jep.15.4.101>.
- Watson, M. W. 1994. Vector autoregressions and cointegration. In Vol. 4 of *Handbook of Econometrics*, ed. R. F. Engle and D. L. McFadden. Amsterdam: Elsevier. [https://doi.org/10.1016/S1573-4412\(05\)80016-9](https://doi.org/10.1016/S1573-4412(05)80016-9).

## Also see

- [TS] **var postestimation** — Postestimation tools for var
- [TS] **dfactor** — Dynamic-factor models
- [TS] **forecast** — Econometric model forecasting
- [TS] **mgarch** — Multivariate GARCH models
- [TS] **sspace** — State-space models
- [TS] **tsset** — Declare data to be time-series data
- [TS] **var intro** — Introduction to vector autoregressive models
- [TS] **var ivsvar** — Instrumental-variables structural vector autoregressive models<sup>+</sup>
- [TS] **var svar** — Structural vector autoregressive models
- [TS] **varbasic** — Fit a simple VAR and graph IRFs or FEVDs
- [TS] **vec** — Vector error-correction models
- [XT] **xtvar** — Panel-data vector autoregressive models<sup>+</sup>
- [U] **20 Estimation and postestimation commands**

*Stata Dynamic Stochastic General Equilibrium Models Reference Manual*

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on citing Stata documentation.